

Máster Universitario "Ciencia y Tecnología Nuclear"

Seminario abierto a estudiantes de otros Másteres o últimos cursos de Grados relacionados con la temática

SEMINARIO AVANZADO:

"Radiactividad, radiaciones y protección radiológica. Aplicaciones con Mathematica"

Prof. José Guillermo Sánchez León.

Noviembre 2017

Parte del material de este *notebook* procede del libro **Mathematica Beyond Mathematics: The Wolfram Language in the Real World** by José Guillermo Sánchez León © Chapman and Hall/CRC; 1 edition (May 9, 2017) (No se permite su reproducción ni distribución).

<https://www.amazon.com/Mathematica-Beyond-Mathematics-Wolfram-Language/dp/149879629X>

La naturaleza esta formada por elementos, cada elemento puede tener uno o mas isótopos, algunos isótopos son inestables y se desintegran. Esta característica naturaleza que nos permite escudriñar sobre el pasado de la Tierra y de nosotros mismos, por medio de la datación radiactiva. Los componentes de los átomos: neutrones y protones, están a su vez constituidos por otras partículas mas básicas: los quarks. A estos y a otros temas relacionados con la Física Nuclear nos referiremos en este capítulo. Para ello haremos amplio uso de las funciones de Mathematica `ElementData`, `IsotopeData` y `ParticleData`.


Introducción a la funciones utilizadas en física nuclear

La siguiente sintaxis la utilizaremos frecuentemente.

```
func["entidad", #] & /@ {"Propiedad1". "Propiedad2"}  
  
{func(entidad, Propiedad1.Propiedad2)}
```

- Debajo se muestra la tabla periódica.

`ColorData["Atoms", "Panel"]`




- La función `ElementData["elemento", "propiedad"]` nos permite obtener información de una determinada propiedad para un elemento dado:

`ElementData["Uranium", "Density"]`

19050. kg/m³

- Los átomos forman moléculas a los que podemos acceder con la función `ChemicalData`.

`ChemicalData["Caffeine", "MoleculePlot"]`



En relación con los isótopos Mathematica incluye la función `IsotopeData["elemento", "propiedad"]`. Especificado el "elemento" podemos escribirlo con el nombre completo en inglés, o preferentemente utilizando su símbolo estándar. Si en "propiedad" escribimos "Symbol" obtendremos los isótopos en notación ^AX.

- Debajo se muestra un ejemplo en el que se pide se muestren los isótopos del carbono (C).

`IsotopeData["C", "Symbol"]`

{⁸C, ⁹C, ¹⁰C, ¹¹C, ¹²C, ¹³C, ¹⁴C, ¹⁵C, ¹⁶C, ¹⁷C, ¹⁸C, ¹⁹C, ²⁰C, ²¹C, ²²C}

- Debajo se muestran los modos de desintegración del uranio 238, los símbolos que indican esas formas de desintegración, la probabilidad con la que se dan y las energías típicas.

```
Grid[IsotopeData["Uranium238", #] & /@ {"DecayModes",
      "DecayModeSymbols", "BranchingRatios", "DecayEnergies"}]
```

AlphaEmission	SpontaneousFission	DoubleBetaDecay
α	SF	$2\beta^-$
1.00	5.45×10^{-7}	2.2×10^{-12}
4269.75 keV	Missing[NotAvailable]	1144.2 keV

Observe que la mayoría de las desintegraciones del U-238 corresponde a emisiones alfa (el valor 1.00 indica que la probabilidad es próxima al 100%). Sin embargo, las fisiones espontáneas (SF), aunque con una probabilidad baja, tienen un papel fundamental en el inicio de reacciones en cadena, pues las SF normalmente conllevan asociadas la emisión de neutrones imprescindibles para iniciar una reacción en cadena.

- Puede utilizar la forma lingüística para obtener la definición de reacción en cadena ("Nuclear chain reaction"). Probablemente cuando escriba "chain reaction definition" le aparecerán varias definiciones. Se requiere la intervención del usuario para que elija entre las opciones que se ofrecen la que se busca.



- 1 noun a series of chemical reactions in which the product of one is a reactant in the next
- 2 noun a self-sustaining nuclear reaction; a series of nuclear fissions in which neutrons released by splitting one atom leads to the splitting of others

- También podemos obtener información por clases (usamos `Shallow` para que no se muestre la salida completa que es muy larga, si quiere ver todas las disponibles elimine `Shallow`. Otra función similar es `Short` para que solo se nos muestre una línea)

```
IsotopeData["Classes"] // Shallow
```

```
{ alpha emission , beta decay , beta delayed alpha emission ,
  beta delayed deuteron emission , beta delayed fission ,
  beta delayed four neutron emission , beta delayed neutron alpha emission ,
  beta delayed neutron emission , beta delayed three neutron emission ,
  beta delayed triton emission , <<37>> }
```

- Cada clase incluye a los isótopos que presentan la característica común definida en la clase. Por ejemplo: si queremos obtener que isótopos que presentan fisión espontánea con emisión beta asociada (un proceso altamente infrecuente, solo observado en 3 isótopos de los más de tres mil disponibles)


```
IsotopeData [ beta delayed fission ISOTOPES ]
{ actinium-230 , protactinium-236 , protactinium-238 }
```

Constante de desintegración, período de semidesintegración, cadena de desintegración

```
Clear ["Global`*"]
```

El caso más sencillo de desintegración radiactiva es el de un isótopo radiactivo padre, A, que se desintegra en un núcleo hijo estable, B, esto es $A \rightarrow B$, donde B es no radiactivo. Este proceso puede representarse por un modelo compartimental simplísimo, como el de la figura. En este ejemplo llamamos $x(t)$ a la cantidad de un determinado isótopo radiactivo A presente en un instante de tiempo t en una muestra (representada por un compartimento).

- Este gráfico se ha elaborado utilizando la herramienta de dibujo (para abrirla pulse Ctrl + D)



La cantidad de átomos (expresada en la unidad que considere apropiada: átomos, gramos, bequerelios, etc.) que se desintegran por unidad de tiempo es proporcional a los existentes en cada momento en la muestra, esto es: $\lambda x(t)$, donde λ es la constante de proporcionalidad, conocida como constante de desintegración radiactiva, específica de cada isótopo.

Supongamos que en $t = 0$ la muestra contiene x_0 . Este proceso podemos representarlo matemáticamente por la sencilla ecuación diferencial de primer orden:

$$\frac{dx}{dt} = -\lambda x(t), \text{ con condición inicial : } x_0 \text{ en } t = 0 \quad (0.1)$$

- La solución de la ecuación anterior podemos obtenerla utilizando DSolve. Observe que utilizamos “=” y no “:=” ¿Por qué? (normalmente es preferible “:=”, pero en este caso optamos por “=” pues así la próxima vez que llame a la función partirá de la solución y no tendrá que resolver de nuevo la ecuación diferencial, en este caso el ahorro de tiempo es despreciable)


```
x1[ t_ , x0_ , λ_ ] =
  x[t] /. DSolve[ {x'[t] == -λ x[t], x[0] == x0}, x[t], t] [[1]]
  e-t λ x0
```

La reescribimos en formato texto

$$x(t) = x_0 e^{-\lambda t} \quad (0.2)$$

- En el siguiente ejemplo dejamos como parámetros el valor de λ y el intervalo superior de t . Además utilizamos en Manipulate una función definida para ello podemos utilizar SaveDefinitions o Initialization.

```
Manipulate[Plot[y[t, λ, 1], {t, 0, t1}, Filling -> Bottom],
  {{λ, 1}, 0, 5}, {{t1, 3}, 1, 10},
  Initialization -> (y[t_, λ_, x0_] := x0 e^{λ (-t)})]
```



En muchas ocasiones en vez de la **constante de desintegración**, λ , se prefiere utilizar el **período de semidesintegración** o **semivida** (*half life* en inglés), $T_{1/2}$, que se define como el tiempo que debe transcurrir para que la actividad x_0 presente en una muestra en $t = 0$ se reduzca a la mitad, es decir a $x_0/2$.

- La relación entre λ y $T_{1/2}$ puede calcularse como sigue

$$\text{cteDes} = \lambda /. \text{Solve}[x1[T, x0, \lambda] == x0/2, \lambda, Reals][[1]]$$

$$\frac{\text{Log}[2]}{T}$$

Otro concepto es el de **vida media** (*mean lifetime* en inglés) que es la esperanza de vida (duración) para un conjunto de isótopos presentes en una muestra. Es lo mismo que calcular la media aritmética de la vida esperada de los isótopos presentes en una muestra.

- Para calcular la media para una variable continua t aplicamos $\langle t \rangle = \int_0^{\infty} t f(t) dt / \int_0^{\infty} f(t) dx$, que en nuestro caso denotamos por τ . Especificamos que $\lambda \in \mathbb{R} > 0$, en otro caso se recibirá un mensaje que le indicará que la solución solo es válida si $\lambda \in \mathbb{R} > 0$.

```

τ = Integrate[t x_0 E^(-t λ), {t, 0, Infinity},
  Assumptions → λ ∈ Reals && λ > 0] / Integrate[x_0 E^(-t λ),
  {t, 0, Infinity}, Assumptions → λ ∈ Reals && λ > 0]

```

$$\frac{1}{\lambda}$$

- Con `IsotopeData` podemos obtener el período de semidesintegración y la vida media del yodo 131, por defecto nos el tiempo en segundo, que podemos convertir a otra unidad utilizando `UnitConvert`.

```

IsotopeData["Iodine131", #] & /@ {"HalfLife", "Lifetime"}
{6.9338 × 105 s, 1.0003 × 106 s}

UnitConvert[
  IsotopeData["Iodine131", #] & /@ {"HalfLife", "Lifetime"}, "days"]
{8.0252 days, 11.578 days}

```

En resumen: la constante de desintegración radiactiva, λ , el período de semidesintegración, $T_{1/2}$, y la vida media, τ , se relacionan como sigue: $T_{1/2} = \text{Log}[2]/\lambda = \tau \text{Log}[2]$, $\tau = 1/\lambda$. **Erróneamente a veces se usa el término vida media en lugar de período de semidesintegración.**

Cadenas de desintegración

```
Clear["Global`*"]
```

Según hemos visto, un isótopo radiactivo es como un padre que tiene un hijo que a su vez puede tener otro, su nieto, y así sucesivamente hasta llegar a un núcleo estable. A veces el hijo es estable, como ocurre con el yodo 131 → xenón 131, pero en otros casos los isótopos radiactivos poseen cadenas de desintegración más complicadas como la desintegración del radio 226 que se muestra en la figura al final de esta sección. Uno de los descendientes del radio 226 es el radón 222 (Rn222), que es el responsable de una parte importante de la radiación natural a la que estamos expuestos todas las personas. Cada vez que respiramos, una parte del aire incluye al gas radón. El contenido de Rn222 varía sustancialmente de unas zonas a otras. Por ejemplo: las casas graníticas poco ventiladas suelen tener un alto contenido en radón. Es así pues el granito contiene radio 226 que al desintegrarse se libera como gas radón.

- Un isótopo se puede desintegrar siguiendo distintas ramas. Por ejemplo: El radio 226 se desintegra con un período de semidesintegración de 5.0×10^{10} s en 3 radisótopos: ^{222}Rn , ^{212}Pb y ^{226}Th . Prácticamente casi todo se desintegra en ^{222}Rn y una fracción insignificante en ^{212}Pb . También puede producirse trazas Th226.

```

ra226 = IsotopeData["Ra226", #] & /@
  {"DaughterNuclides", "BranchingRatios", "HalfLife"};

TableForm[ra226,
  TableHeadings -> {{ "Isótopo", "fracción", "T (s)", None }}]

```

Isótopo	radon-222	lead-212	thorium-226
fracción	1.00	2.6×10^{-11}	Missing[Unknown]
T (s)	5.0×10^{10} s		

- En distintos contextos es probable que al utilizar **IsotopeData** se encuentre que en el resultado aparezca la palabra “Missing” o “___”. Si quiere eliminar estas soluciones resulta muy útil el empleo **DeleteMissing** que elimina todos los elementos de la lista cuya cabecera empieza por h, como se muestra en el ejemplo .

```
is1 = IsotopeData["U238", "DaughterNuclides"]
```

```
{thorium-234, —, plutonium-238}
```

```
DeleteMissing[is1]
```

```
{thorium-234, plutonium-238}
```

- La siguiente función permite elegir la rama principal, y dentro de ésta podemos quedarnos con el número de descendientes n que nos interese. Observe el empleo de *iso_String*, y *n_Integer* con los que forzamos a que se utilice como formato de entrada: “*nombre del isótopo*” (entre comillas), y n entero (número de descendientes que se desea mostrar); en otro caso la función no se ejecutará.

```
cadprin[iso_String, n_Integer] :=
```

```
NestList[First[IsotopeData[#, "DaughterNuclides"]] &, iso, n]
```

- La aplicamos al Ra-226, nos quedamos con los 3 primeros descendientes

```
cadprin["Ra226", 3]
```

```
{Ra226, radon-222, polonium-218, lead-214}
```

Aplicación al iodo 131

Las emisiones de iodo 131 procedentes de la central Fukushima Daichi, a causa de un accidente producido por un tsunami en marzo de 2011, se estimaban a finales de abril 2011 de 1.5×10^{17} Bq (<http://www.nisa.meti.go.jp/english/files/en20110412-4.pdf>). Nos planteamos cuál habrá sido la cantidad emitida en masa y cómo evolucionará con el tiempo.

La actividad la podemos convertir en masa teniendo en cuenta la relación $A = \lambda n_A = \lambda m N_A / m_a \rightarrow m = m_a A / (\lambda N_A)$.

- El iodo 131 se desintegra en xenón 131, isótopo estable que a temperatura ambiente es un gas. Utilizamos **LayeredGraphPlot** para representar la cadena de desintegración.

```
LayeredGraphPlot[{IsotopeData["I131", "Symbol"] →  
IsotopeData[IsotopeData["I131", "DaughterNuclides"]][[1]],  
"Symbol"]}, Left, VertexLabeling → True]
```



- Necesitamos conocer la constante de desintegración, λ_{131} , y la masa atómica que podemos obtener con **IsotopeData[]** obtenemos la masa atómica (aunque podríamos emplear con muy buena aproximación 131) y la vida media τ (que está dado en seg.). Entonces calculamos $\lambda = 1/\tau$.

```
{mat131, λ131} = {IsotopeData["Iodine131", "AtomicMass"],  
1/IsotopeData["Iodine131", "Lifetime"]}
```

```
{130.906124609 u, 9.9967 × 10-7 per second}
```

- Ahora aplicamos $m = m_a A / (\lambda N_A)$ donde N_A (número de Avogadro) podemos obtenerlo por medio de **Mathematica** utilizando la notación lingüística “Avogadro Number”) o con la función **Quantity**. **UnitConvert** muestra la unidad en el sistema internacional, en este caso es un número sin unidad.

```
Quantity["Avogadro Number"]
1 NA
NA = UnitConvert[1 NA]
6.022141 × 1023
1.5 × 1017
QuantityMagnitude[mat131] / (QuantityMagnitude[λ131] * NA) "gramos"
32.617 gramos
```

- Nos planteamos calcular el tiempo en el que la actividad del yodo 131 se habrá reducido al 0.1% de la inicial. El resultado lo expresamos en días.

```
UnitConvert[t /. Solve[x1[t, 100, λ131] == 0.1, t], "days"]
{79.9774 days}
```

Energía de enlace

La datación de nuestro pasado

Partículas subatómicas. Quarks.

Desde principios del siglo XX sabemos que los átomos están a su vez compuestos por otras partículas. Inicialmente se pensó que las partículas elementales eran los protones, neutrones y electrones, serían los auténticos átomos en el sentido que serían indivisibles, pero al menos los neutrones y protones no lo son. Para obtener información de partículas subatómicas podemos utilizar la función: `ParticleData`.

- Un neutrón libre (no ligado a un átomo) tiene una semivida de pocos minutos:


```
ParticleData["Neutron", "Lifetime"]
885.6 s
```

- Esto se explica debido a que el neutrón, al igual que el protón, no es realmente una partícula elemental, sino que están formados por dos tipos de quarks:

```
ParticleData[#, "QuarkContent"] & /@ {"Proton", "Neutron"}
{{{d, u, u}}, {{d, d, u}}}
```

- El mejor modelo que disponemos en la actualidad para explicar las partículas y fuerzas (excluida la gravedad) es lo que se conoce como "Modelo estándar de la física de partículas"


```
WikipediaData["Standard_Model", "ImageList"][[1]]
```



```
ParticleData["Lepton"]
```

```
{  $\nu_e$ ,  $\bar{\nu}_e$ ,  $\nu_\mu$ ,  $\bar{\nu}_\mu$ ,  $\nu_\tau$ ,  $\bar{\nu}_\tau$ ,  $e^-$ ,  $e^+$ ,  $\mu^-$ ,  $\mu^+$ ,  $\tau^-$ ,  $\tau^+$  }
```

```
ParticleData[#, "FullSymbol"] & /@ %
```

```
{  $\nu_e$ ,  $\bar{\nu}_e$ ,  $\nu_\mu$ ,  $\bar{\nu}_\mu$ ,  $\nu_\tau$ ,  $\bar{\nu}_\tau$ ,  $e^-$ ,  $e^+$ ,  $\mu^-$ ,  $\mu^+$ ,  $\tau^-$ ,  $\tau^+$  }
```

```
ParticleData[#, "FullSymbol"] & /@ ParticleData["Quark"]
```

```
{u,  $\bar{u}$ , d,  $\bar{d}$ , s,  $\bar{s}$ , c,  $\bar{c}$ , b,  $\bar{b}$ , t,  $\bar{t}$ }
```

```
ParticleData[#, "FullSymbol"] & /@ ParticleData["Gluon"]
```

```
Entity[$Failed, g]
```

- Para obtener información de alguna de las características de estas partículas como sigue (como ejemplo utilizamos los quark arriba (up) y abajo (down), que hemos visto son los que forman los protones y neutrones):

```
vals =
```

```
Table[ParticleData[#, prop], {prop, {"Symbol", "Charge", "Mass"}}] & /@ {"DownQuark", "UpQuark"}
```

```
{{d,  $-\frac{1}{3}$ , 5.0 MeV/c2}, {u,  $\frac{2}{3}$ , 2.2 MeV/c2}}
```

```
Text[Grid[Prepend[vals, {"", "carga", "masa(MeV/c2)"}], Frame → All,
  Background → {None, {{{LightBlue, White}}, {1 → LightYellow}}}]
```

	carga	masa(MeV/c ²)
d	$-\frac{1}{3}$	5.0 MeV/c ²
u	$\frac{2}{3}$	2.2 MeV/c ²

- Nótese la pequeñísima masa de los quarks comparada con la de un neutrón (dos quark d y un quark u) o un protón (dos quark u y un quark d):

```
{qd, qu} = ParticleData[#, "Mass"] & /@ {"DownQuark", "UpQuark"}
```

```
{5.0 MeV/c2, 2.2 MeV/c2}
```

```
{quarksproton, quarkneutron} =
```

```
100 { (2 qu + 2 × 1 qd) / ParticleData["Proton", "Mass"],
  (2 qu + 1 qd) / ParticleData["Neutron", "Mass"] } "%"
```

```
{1.5 %, 1.0 %}
```

Es decir: los quarks suponen solo entre el 1 y el 1.5% la masa de los núcleos atómicos, el resto corresponden a la interacción entre ellos, por eso no hay forma de que se presenten de forma libre.

La masa de los quarks y de otras partículas, como el electrón, es atribuida al campo de Higgs (o mejor el BEH) cuya existencia fue confirmada con la detección del bosón de Higgs, oficialmente anunciada por el CERN el 4 de julio de 2012. Entonces, ¿ nuestra masa viene del campo de Higgs? Solo una pequeña parte:

Según hemos visto los quarks suponen solo entre el 1% y el 1.5% de la masa de los núcleos, la masa de los quarks es la que es explicada por el campo de Higgs. Por tanto, para una persona que pese 80 kg el campo de Higgs explica aproximadamente 1 kg. ¿De dónde procede la diferencia? La aportan los gluones, g, que mantienen unidos a los quarks y los fotones, en este caso debemos hablar de masa equivalente $m = E/c^2$.

Estas partículas no interaccionan con el campo de Higgs o BEH pero sí con el campo gravitatorio. Sin embargo, si no existiese este campo, los electrones irían a la velocidad de la luz y no se habrían podido formar los átomos y por tanto no existiríamos.

Recursos adicionales